Electron-positron pair production in the collision of real photon beams with wide energy distributions

L Esnault *et al* 2021 Plasma Phys. Control. Fusion **63** 125015
Notebook: Óscar Amaro, October 2022 @ GoLP-EPP
Contact: oscar.amaro@tecnico.ulisboa.pt

Introduction

In this notebook we reproduce Figures 3, 4 and 5

Figure 3

In[•]:=
```
Clear[m, c, s, re, σγγ]
c = 1;
m = 0.511;
re = 1;
σγγ =
 4 π re^2 m^2 c^4/s ((2 + 8 m^2 c^4/s - 16 m^4 c^8/s^2) Log[(Sqrt[s] + Sqrt[s - 4 m^2 c^4])/(2 m c^2)] -
    Sqrt[1 - 4 m^2 c^4/s] (1 + 4 m^2 c^4/s))
LogLinearPlot[σγγ /. {s → sqrts^2}, {sqrts, 10^-1, 10^2},
 PlotRange → {{10^-1, 10^2}, {0, 2.5}}, Frame → True,
 FrameLabel → {"√s[MeV]", "σγγ[re^2]"}]
```

Out[•]=
$$\frac{1}{s}\,3.28134\left(-\sqrt{1-\frac{1.04448}{s}}\left(1+\frac{1.04448}{s}\right)+\left(2-\frac{1.09095}{s^2}+\frac{2.08897}{s}\right)\text{Log}\left[0.978474\left(\sqrt{-1.04448+s}+\sqrt{s}\right)\right]\right)$$

Out[•]=




```
(* peak of cross section occurs at s=2.06 MeV^2*)
FindMaximum[σγγ, {s, 1.1}]
```

Out[•]= {2.14164, {s → 2.05544}}

```
(* peak of cross section/10 occurs at s=71.7 MeV^2,
as mentioned in caption of Figure 6*)
FindRoot[σγγ - 2.1416410079961308` / 10, {s, 2}]
```

Out[•]= $\left\{s \to 71.7428 + 2.49495 \times 10^{-19}\,i\right\}$

Figure 4

```
In[•]:= LogLogPlot[
  {Exp[-EiKi], 1/Gamma[0.05] (EiKi)^(-0.95) Exp[-EiKi], HeavisideTheta[1 - EiKi]},
  {EiKi, 10^-3, 10^1}, PlotLegends → {"Brem", "LMC", "IC"},
  Frame → True, FrameLabel → {"Ei/Ki", "function"}, PlotPoints → 10,
  Filling → None, PlotStyle → {Directive[Blue, Dashed],
    Directive[Red, DotDashed], Directive[Purple]}, GridLines → Automatic]
```

Out[•]=




```
(* the "theoretical" distributions are already normalized *)
Clear[EiKi]
Integrate[Exp[-EiKi], {EiKi, 0, ∞}]
Integrate[1/Gamma[0.05] (EiKi)^(-0.95) Exp[-EiKi], {EiKi, 0, ∞}]
Integrate[HeavisideTheta[1 - EiKi], {EiKi, 0, ∞}]
```

Out[•]= 1

Out[•]= 1.

Out[•]= 1

Figure 5: numerical integrals and fitted functions

```
Clear[h, ℰ, a, b, n, m, re]
Clear[m, c, s, re, σγγ, smin]
Clear[gBremBrem, gLMCLMC, gICIC, σ11BremBrem, σ11LMCLMC,
 σ11ICIC, intgBremBrem, intgLMCLMC, intgICIC, logsqrtℰmin, dlogsqrtℰ]
c = 1;
m = 0.511;
re = 1;
```

```
σγγ[s_?NumericQ] := σγγ[s] =
  4 π re^2 m^2 c^4/s ((2 + 8 m^2 c^4/s - 16 m^4 c^8/s^2) Log[(Sqrt[s] + Sqrt[s - 4 m^2 c^4])/(2 m c^2)] -
     Sqrt[1 - 4 m^2 c^4/s] (1 + 4 m^2 c^4/s))

(* we can only integrate for s>4m^2c^4 *)
smin = 4 m^2 c^4;

gBremBrem[EiKi_?NumericQ] := gBremBrem[EiKi] = Exp[-EiKi] // Quiet
gLMCLMC[EiKi_?NumericQ] :=
 gLMCLMC[EiKi] = 1/Gamma[0.05] (EiKi)^(-0.95) Exp[-EiKi] // Quiet
gICIC[EiKi_?NumericQ] := gICIC[EiKi] = HeavisideTheta[1 - EiKi] // Quiet

intgBremBrem[η_?NumericQ] :=
 intgBremBrem[η] = NIntegrate[gBremBrem[s/η] × σγγ[s], {s, smin, ∞}] // Quiet

σ11BremBrem[ξ_] :=
 1/ξ NIntegrate[gBremBrem[η/ξ]/η intgBremBrem[η], {η, 0, ∞}] // Quiet

intgLMCLMC[η_?NumericQ] :=
 intgLMCLMC[η] = NIntegrate[gLMCLMC[s/η] × σγγ[s], {s, smin, ∞}] // Quiet

σ11LMCLMC[ξ_] := 1/ξ NIntegrate[gLMCLMC[η/ξ]/η intgLMCLMC[η], {η, 0, ∞}] // Quiet

intgICIC[η_?NumericQ] :=
 intgICIC[η] = NIntegrate[gICIC[s/η] × σγγ[s], {s, smin, ∞}] // Quiet

σ11ICIC[ξ_] := 1/ξ NIntegrate[gICIC[η/ξ]/η intgICIC[η], {η, 0, ∞}] // Quiet

(* generate data for figure 5 *)
(* logsqrtξ=Log10[Sqrt[ξ]] → ξ=(10^logsqrtξ)^2*)
logsqrtξmin = -0.9;
dlogsqrtξ = 0.03;
lstσ11BremBrem = ParallelTable[{(10^logsqrtξ), σ11BremBrem[(10^logsqrtξ)^2]},
   {logsqrtξ, logsqrtξmin, 5, dlogsqrtξ}];
lstσ11LMCLMC = ParallelTable[{(10^logsqrtξ), σ11LMCLMC[(10^logsqrtξ)^2]},
   {logsqrtξ, logsqrtξmin, 5, dlogsqrtξ}];
lstσ11ICIC = ParallelTable[{(10^logsqrtξ), σ11ICIC[(10^logsqrtξ)^2]},
   {logsqrtξ, logsqrtξmin, 5, dlogsqrtξ}];
```

Plot generated data and fits (from parameters in paper, not from the data generated here)

In[•]:=
```
Clear[h, ζ, a, b, n, m, re]
Clear[m, c, s, re, σγγ]
c = 1;
m = 0.511;
re = 1;
σγγ =
  4 π re^2 m^2 c^4/s ((2 + 8 m^2 c^4/s - 16 m^4 c^8/s^2) Log[(Sqrt[s] + Sqrt[s - 4 m^2 c^4])/(2 m c^2)] -
    Sqrt[1 - 4 m^2 c^4/s] (1 + 4 m^2 c^4/s));
h[ζ_, a_, b_, n_, m_] := re^2 (a / ζ^n) Exp[-b / ζ^m]
Show[{LogLogPlot[{σγγ /. {s → sqrtζ^2}, h[sqrtζ^2, 23.3, 3.94, 0.674, 0.374],
    h[sqrtζ^2, 9.82 × 10^-2, 4.13, 3.32 × 10^-3, 0.221],
    h[sqrtζ^2, 10.5, 6.01, 0.552, 0.798]}, {sqrtζ, 10^-1, 10^6},
   PlotStyle → {Directive[Black, Dashed], Directive[Blue], Directive[Red],
     Directive[Purple]}, PlotRange → {{10^-1, 10^6}, {10^-4, 10^1}},
   PlotLegends → {"Mono-Mono", "Brem-Brem", "LMC-LMC", "IC-IC"},
   GridLines → Automatic, Frame → True,
   FrameLabel → {"√ζ [MeV]", "σintγγ[re^2]"}],
  ListLogLogPlot[{lstσ11BremBrem, lstσ11LMCLMC, lstσ11ICIC},
   PlotStyle → {Directive[Blue, PointSize[0.01]],
     Directive[Red, PointSize[0.01]], Directive[Purple, PointSize[0.01]]},
   PlotRange → {{10^-1, 10^6}, {10^-4, 10^1}}]}]
LogLinearPlot[{σγγ /. {s → sqrtζ^2}, h[sqrtζ^2, 23.3, 3.94, 0.674, 0.374],
  h[sqrtζ^2, 9.82 × 10^-2, 4.13, 3.32 × 10^-3, 0.221],
  h[sqrtζ^2, 10.5, 6.01, 0.552, 0.798]}, {sqrtζ, 10^-1, 10^6}, PlotStyle →
  {Directive[Black, Dashed], Directive[Blue], Directive[Red], Directive[Purple]},
 PlotRange → {{10^-1, 10^2}, {10^-4, 2.2}},
 PlotLegends → {"Mono-Mono", "Brem-Brem", "LMC-LMC", "IC-IC"},
 GridLines → Automatic, Frame → True, FrameLabel → {"√ζ [MeV]", "σintγγ[re^2]"}]
```

Out[•]=



Out[•]=


σintyy[re^2]
2.0
1.5
1.0
0.5
0.1
0.5
1
5
10
50
100
√ζ [MeV]
Mono–Mono
Brem–Brem
LMC–LMC
IC–IC